


Kabelbelegung für Verbindungskabel zum Interface 6050/6051

Auf PC-Seite wird eine 9- oder 25-polige D-SUB-Kupplung verwendet. Auf der anderen Seite wird der dem Interface 6050 beiliegende 6-polige DIN-Rundstecker verwendet. Ab dem Interface 6051 ist dieses Kabel bereits im Lieferumfang enthalten. Hat Ihr PC einen 25-poligen Stecker, dann können Sie einfach einen handelsüblichen 9-auf-25-Adapter einsetzen.



Im folgenden sind drei kleine Programme abgedruckt, die zum Test der Verbindung zum Interface eingesetzt werden können. Beim Arbeiten im DOS-Fenster (ohne Windows zu beenden) ist häufig keine Verbindung zum Interface möglich. Deshalb empfehlen wir Windows zu beenden, und das Programm QBASIC.EXE im DOS-MODUS starten. Dort können Sie dann diese Programme eingeben.

Bitte denken Sie daran, den Text in Zeile 100 von **COM1:2400** auf **COM2:2400** zu ändern, wenn Sie das Interface an die zweite Schnittstelle (COM2) angeschlossen haben.

DIGI1 – Dieses Programm kann genutzt werden, um die Verbindung zum Interface herzustellen. Zum Test der Verbindung empfehlen sich die Befehle STOP und GO. Bitte beachten Sie, daß das Interface nur Loks steuern kann, die nicht gerade von einem Fahrgerät (von Hand) gesteuert werden. Sie müssen deswegen am Fahrgerät eine andere gültige Loknummer eingeben.

```

100 OPEN "COM1:2400,N,8,2,CS10000,DS" FOR RANDOM AS #1
110 CLS
200 INPUT " Byte1 (Geschw. 0-15) "; GE
210 INPUT " Byte2 (Loknr. 1-80) "; LN
290 IF LOF(1) < 512 THEN GOTO 290
300 PRINT #1, CHR$(GE); CHR$(LN);
310 GOTO 200
    
```

Das Programm DIGI1 kann folgende Befehle an das Interface senden:

Byte1	Byte2	Befehl	Erklärung
96	96	GO	
97	97	STOP	
0	1...80	Lok anhalten	Fahrbefehl mit Geschwindigkeit 0, Licht aus
1...14	1...80	Lok fahren	Fahrbefehl mit Licht aus
15	1...80	Fahrtrichtung umkehren	Lok bleibt stehen, und wechselt die Fahrtrichtung, Licht aus
16	1...80	Lok anhalten	Geschwindigkeit 0 mit Licht an
17...30	1...80	Lok fahren	Fahrbefehle mit Licht an
31	1...80	Fahrtrichtung umkehren	Lok bleibt stehen, wechselt die Fahrtrichtung, und hat Licht an
64	1...80	alle Funktionen aus	F4, F3, F2, F1
65	1...80	Funktion F1 ein	F4, F3, F2, F1
66	1...80	Funktion F1 ein	F4, F3, F2 , F1
67...	1...80	Funktion F1 und F2 ein	F4, F3, F2 , F1
79	1...80	alle Funktionen ein	F4, F3, F2, F1

Kabelbelegung für Verbindungskabel zum Interface 6050/6051

DIGI2 – Dieses Programm kann Weichen und Signale stellen. Zuerst wird die Nummer der Weichen/Signale abgefragt, dann kann mit den Tasten R und G geschaltet werden. Die Schaltdauer beträgt dabei 0,2 Sekunden.

```
100 OPEN "COM1:2400,N,8,2,CS10000,DS" FOR RANDOM AS #1
110 CLS : ROT = 34 : GRUEN = 33 : AUS = 32 : TW = 1
200 INPUT " Magnetartikel (1-255) "; MN
210 INPUT " Rot oder Gruen (R/G) "; MS$
220 IF MS$ = "r" OR MS$ = "R" THEN MS = ROT: GOTO 300
230 IF MS$ = "g" OR MS$ = "G" THEN MS = GRUEN: GOTO 300
240 GOTO 200
300 IF LOF(1) < 512 THEN GOTO 300
310 PRINT #1, CHR$(MS); CHR$(MN);
311 T = TIMER +0.2
320 IF TIMER < T THEN 320
325 IF LOF(1) < 512 THEN GOTO 325
330 PRINT #1, CHR$(AUS);
340 GOTO 200
```

DIGI3 – Dieses Programm liest Rückmeldemodule, und stellt deren Kontakte am Bildschirm dar. Eine 1 bedeutet Kontakt belegt, eine 0 bedeutet Kontakt frei. Wenn kein Rückmeldemodul am Interface angeschlossen ist, wird das Programm jeweils 0 anzeigen.

```
100 OPEN "COM1:2400,N,8,2,CS10000,DS" FOR RANDOM AS #1
110 CLS : RM = 1
200 LOCATE 1, 29, 1: PRINT "      ": LOCATE 1, 1, 1
210 INPUT " Rueckmeldemodulnr. (1-63) "; RM
220 IF RM < 1 OR RM > 63 THEN GOTO 200
230 IF LOF(1) < 512 THEN GOTO 230
240 PRINT #1, CHR$(RM + 192);
250 PRINT : PRINT
260 PRINT "Schalter:  1  2  3  4  5  6  7  8": PRINT "ein/aus :  ";
270   A = ASC(INPUT$(1, #1))
280 FOR I = 7 TO 0 STEP -1 : PRINT SGN(A AND 2 ^ I); : NEXT
310 PRINT : PRINT : PRINT
320 PRINT "Schalter:  9 10 11 12 13 14 15 16": PRINT "ein/aus :  ";
330   B = ASC(INPUT$(1, #1))
340 FOR I = 7 TO 0 STEP -1 : PRINT SGN(B AND 2 ^ I); : NEXT
380 GOTO 200
```

Tips zur Fehlersuche:

- Gehen Sie schrittweise vor:
 1. Mit DIGI1 sollten die Befehle STOP/GO die Kontrolllampe der 6021 aus/ein-schalten (→Kabel ist OK)
 2. Mit DIGI3 sollten bei Modulnr 31 nur 16 mal der Kontakt mit 0 gemeldet werden (→Kabel ist OK)
 3. Mit DIGI2 sollten die Weichen und Signale schaltbar sein(→Interface nach 1987 OK)
 4. Mit DIGI1 sollten Loks frei steuerbar sein(→Anwender ist auch OK;-)
- Manche Laptops können nicht zum Interface verbinden, weil ihre RS232(COM1) nicht normgerecht sendet.
- Wenn alle Rückmeldekontakte besetzt anzeigen ist der S88 verkehrt angeschlossen.
- Im DOS-Fenster von Windows 95/98 ist oft keine Verbindung zum Interface möglich.
- Sie können QBASIC.EXE mit der Funktion Extras/Suchen /Dateien im Explorer suchen.
- Wenn QBASIC.EXE nicht auf der Festplatte ist, finden Sie es auf der Original Windows-CD.
- Wenn DIGI1 mit Fehler 24: „Device Timeout“ abbricht, ist vielleicht die falsche COM-Schnittstelle eingetragen. versuchen Sie mal COM1 gegen COM2 zu tauschen. (Kabel oder Programm oder beides)
- Gehen Weichen (DIGI2) aber keine Loks (DIGI1), dann ist meist im Fahrgerät dieselbe Loknr eingetippt.
- Gehen STOP/GO, aber keine Weichen und keine Loks, könnte es ein Interface von 1985 sein.
- Ein Interface von 1985 kann mit 2400,N,8,1,CS0,DS zur Mitarbeit überredet werden, wird aber von Märklin auch kostenlos auf den fehlerfreien Stand von 1987 korrigiert. (Software-update)